NAG Library Routine Document

G05TKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05TKF generates a vector of pseudorandom integers, each from a discrete Poisson distribution with differing parameter.

2 Specification

```
SUBROUTINE G05TKF(M, VLAMDA, STATE, X, IFAIL)
INTEGER             M, STATE(lstate), X(M), IFAIL
double precision    VLAMDA(M)
```

3 Description

G05TKF generates m integers x_j, each from a discrete Poisson distribution with mean λ_j, where the probability of $x_j = I$ is

$$P(x_j = I) = \frac{\lambda_j^I \times e^{-\lambda_j}}{I!}, \qquad I = 0, 1, \ldots,$$

where

$$\lambda_j \geq 0, \qquad j = 1, 2, \ldots, m.$$

The methods used by this routine have low set up times and are designed for efficient use when the value of the parameter λ changes during the simulation. For large samples from a distribution with fixed λ using G05TJF to set up and use a reference vector may be more efficient.

When $\lambda < 7.5$ the product of uniforms method is used, see for example Dagpunar (1988). For larger values of λ an envelope rejection method is used with a target distribution:

$$f(x) = \tfrac{1}{3} \qquad \text{if } |x| \leq 1,$$

$$f(x) = \tfrac{1}{3}|x|^{-3} \quad \text{otherwise.}$$

This distribution is generated using a ratio of uniforms method. A similar approach has also been suggested by Ahrens and Dieter (1989). The basic method is combined with quick acceptance and rejection tests given by Maclaren (1990). For values of $\lambda \geq 87$ Stirling's approximation is used in the computation of the Poisson distribution function, otherwise tables of factorials are used as suggested by Maclaren (1990).

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05TKF.

4 References

Ahrens J H and Dieter U (1989) A convenient sampling method with bounded computation times for Poisson distributions *Amer. J. Math. Management Sci.* 1–13

Dagpunar J (1988) *Principles of Random Variate Generation* Oxford University Press

Maclaren N M (1990) A Poisson random number generator *Personal Communication*

5 Parameters

1: M – INTEGER *Input*

On entry: m, the number of Poisson distributions for which pseudorandom variates are required.

Constraint: $\mathrm{M} \geq 1$.

2: VLAMDA(M) – ***double precision*** array *Input*

On entry: the means, λ_j, for $j = 1, 2, \ldots, \mathrm{M}$, of the Poisson distributions.

Constraint: $0.0 \leq \mathrm{VLAMDA}(j) \leq \mathrm{X02BBF}/2$, for $j = 1, 2, \ldots, \mathrm{M}$.

3: STATE($lstate$) – INTEGER array *Communication Array*

Note: $lstate =$ LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF.

On entry: contains information on the selected base generator and its current state.

On exit: contains updated information on the state of the generator.

4: X(M) – INTEGER array *Output*

On exit: the m pseudorandom numbers from the specified Poisson distributions.

5: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, $\mathrm{M} < 1$.

IFAIL $= 2$

On entry, $\mathrm{VLAMDA}(j) < 0.0$ for at least one value of j.

On entry, $2 \times \mathrm{VLAMDA}(j) > \mathrm{X02BBF}()$ for at least one value of j.

IFAIL $= 3$

On entry, STATE vector was not initialized or has been corrupted.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example prints ten pseudorandom integers from five Poisson distributions with means $\lambda_1 = 0.5$, $\lambda_2 = 5$, $\lambda_3 = 10$, $\lambda_4 = 500$ and $\lambda_5 = 1000$. These are generated by ten calls to G05TKF, after initialization by G05KFF.

9.1 Program Text

```
*     G05TKF Example Program Text
*     Mark 22 Release. NAG Copyright 2006.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          MSEED, MSTATE, M, N
      PARAMETER        (MSEED=1,MSTATE=633,M=5,N=10)
*     .. Local Scalars ..
      INTEGER          GENID, I, IFAIL, J, LSEED, LSTATE, SUBID
*     .. Local Arrays ..
      DOUBLE PRECISION LAMBDA(M)
      INTEGER          SEED(MSEED), STATE(MSTATE), X(M)
*     .. External Subroutines ..
      EXTERNAL         G05KFF, G05TKF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05TKF Example Program Results'
      WRITE (NOUT,*)
*     Set the distribution parameter LAMBDA
      LAMBDA(1) = 0.5D0
      LAMBDA(2) = 5.0D0
      LAMBDA(3) = 1.0D1
      LAMBDA(4) = 5.0D2
      LAMBDA(5) = 1.0D3
*     Initialize the seed
      SEED(1) = 1762543
*     GENID and SUBID identify the base generator
      GENID = 1
      SUBID = 1
*     Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 40
      END IF
*     Generate N sets of the M variates
      DO 20 I = 1, N
         IFAIL = 1
         CALL G05TKF(M,LAMBDA,STATE,X,IFAIL)
         IF (IFAIL.NE.0) THEN
            WRITE (NOUT,99997) IFAIL
            GO TO 40
         END IF
*        Display the variates
         WRITE (NOUT,99999) I, (X(J),J=1,M)
   20 CONTINUE
*
   40 CONTINUE
*
99999 FORMAT (1X,6(1X,I12))
99998 FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997 FORMAT (1X,' ** G05TKF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05TKF Example Program Results

          1           1           6          12         507        1003
          2           0           9          11         520        1028
          3           1           3           7         483        1041
          4           0           3          11         513        1012
          5           1           5           9         496         940
          6           0           6          17         548         990
          7           1           9           8         512        1035
          8           0           4          10         458        1029
          9           1           6          13         523         971
         10           0           9          16         519         999
```